

In this issue we will examine

- 1) trusses
- 2) wood members in combined axial and bending stress

Using the programs provided herein, and in previous issues, we will design 1) a wood truss, and 2) wood columns, beams and joists, in a pole building.

TRUSSES




In designing a truss, the truss configuration must be selected, that is, the pattern of truss members.

Commonly used truss patterns bear names such as Fink, Howe, Kingpost, Warren, Belgian, and Pratt.

Their configurations are:

Triangular
("Pitched")

Fink




Parallel
Chord
("Flat")


Fink configuration is not applicable to parallel chord trusses.

Triangular
("Pitched")

Howe




Parallel
Chord
("Flat")



In the Howe truss the vertical web members are in tension under ordinary loading conditions; diagonal members are in compression. The tension members are thus shorter than the compression members. This is disadvantageous because compression members should ideally be as short as possible.


Kingpost



Kingpost configuration is not applicable to parallel chord trusses.

Warren


Warren configuration is not applicable to triangular trusses.



In Warren trusses, chord members are generally of equal length; also web members are of equal length. For ordinary loading conditions the web members are in alternate compression-tension-compression-tension, etc. Note that the Warren truss pattern is commonly used in design of steel bar joists.



Triangular
("Pitched")

Belgian



In the Belgian truss, all compression web members are perpendicular to the top chord. Note that all compression web members are shorter than the tension members, and all web members extend from top to bottom chord.

Pratt



Under ordinary loading conditions the compression members in the Pratt truss are vertical and are shorter than the tension members, which overcomes the disadvantage inherent in the Howe Truss. The Pratt pattern is commonly used in parallel chord trusses of wood construction.

Parallel
Chord
("Flat")

Belgian configuration
is not applicable to
parallel chord trusses.

PROGRAM - PRATT TRUSS

```
20 "A" LPRINT "***PRATT TRUSS***":
  LPRINT
25 LPRINT " BY H.C. HALL, PE":
  LPRINT
30 CLEAR: DIM C(15), X(20), Y(20),
  T(160), GAM(20), W(32)
45 TEXT: LPRINT: DEGREE: WAIT 200
```

An optional set of instructions is provided.

```
47 A$="N": BEEP 4: INPUT "DO U WANT
  INSTRUCTS? (Y/N)"; A$
48 IF A$="N" THEN 83
49 LPRINT " PGM SOLVES PRATT
  TRUSS: MEMBER GEOM/LGTH/FORCES;
  DRAWS FB DIAG TO SCALE"
50 LPRINT : LPRINT "X AXIS: BOT
  CHORD": LPRINT "Y AXIS: LEFT
  REACT": LPRINT
55 LPRINT "TOP & BOT CHORD NEED
  NOT BE PARALLEL": LPRINT
60 LPRINT "INPUT TRUSS DIMNS BUT
  BE CONSISTENT"
65 LPRINT "FOR EXAMPLE, INPUT
  FEET OR INCHES"
70 LPRINT "NOT BOTH": LPRINT
73 LPRINT "DESIGNATE": LPRINT "NO.
  OF PANELS": LPRINT "COUNTING
  FROM LEFT WHICH ARE TO HAVE"
76 LPRINT "UP-LEFT, DOWN-RIGHT
  WEB MEMBERS"
83 PRINT USING; "NO. OF PANLS WITH
  UP-LEFT"
84 PRINT "WEB DIAGS (FROM LEFT)"
```

Program requires user to input truss geometry/dimensions.

```
85 INPUT "NO. OF UP-LEFT PANLS=";
  J: LPRINT J: JP = J: LPRINT
86 JJ=2*J+1
87 IF J<1 LPRINT "ERROR-NEED AT
  LEAST ONE UP-LEFT PANEL": GO
  TO 85
88: INPUT "SPAN="; S: S=INT (1000 *S
  +.5)/1000: LPRINT USING "#####.###";
  S: LPRINT
90 INPUT "DEPTH, LEFT END="; A:
  LPRINT A
100 INPUT "DEPTH, RIGHT END="; B:
  LPRINT B: LPRINT
```

```

115 INPUT "NO. OF PANELS=";M:
    USING:LPRINT M:M=M-1:LPRINT
117 IF J>M LPRINT "ERROR:TOO
    MANY UP-LEFT PANLS":GOTO
    85
120 X(1)=0:USING
130 BEEP 4:PRINT "INPUT PANEL
    WIDTHS":PRINT "PANELS 1 THRU
    ";M
135 FOR I=1 TO M:N=I+1:INPUT C(I):
    X(N)=X(I)+C(I):NEXT I
140 FOR I=1 TO M:LPRINT USING
    "#####.##"; C(I):NEXT I
142 KK=S-X(N):KK=INT (100*KK+.5)/100
143 LPRINT KK:LPRINT "- - - - -
    - - - - - ":LPRINT"
    ";S
145 N=M+1:LF 2

```

The angle of the top chord with the horizontal is computed as variable "PHI". Although PC-2 accomodates three letter variables, the variable is actually assigned to "PH".

```

155 PHI=ATN ((B-A)/S): PHI=INT
    (PHI*100+.5)/100
160 FOR I=1 TO N

```

Program calculates lengths of vertical web members.

```

170 Y(I)=A+X(I)*(B-A)/S:Y(I)=INT (100*
    Y(I)+.5)/100
180 NEXT I
185 X(M+2)=S:Y(M+2)=B
186 LPRINT "ANGLE OF TOP CHORD
    W/HORZ= ";PHI:LPRINT"
    DEG"

```

Program prints out truss dimensions at user option.

```

187 B$="N":BEEP 4:INPUT "DOCUMENT
    DIMNS? (Y/N)";B$
190 IF B$="N" THEN 202
192 LPRINT "SPAN= ";S:LPRINT:
    LPRINT "TRUSS DEPTH AT LEFT
    END=";A:LPRINT
193 LPRINT "TRUSS DEPTH AT RIGHT
    END=":LPRINT B

```

```

198 LF 4:LPRINT "***PANEL DIMNS***:
    LPRINT

```

Program calculates member dimensions in subroutine 5000.

```

202 FOR I=1 TO J:U=Y(I):V=(X(I+1)-
    X(I)):GOSUB 5000
203 NEXT I
205 FOR I=(J+1) TO (M+1):U=Y(I+1):
    V=(X(I+1)-X(I)):GOSUB 5000
206 NEXT I

```

Lines 208 and 209 define the scaling factor, D, such that when the free body diagram is drawn, the average truss depth will not exceed 45 units, and the truss length will not exceed 400 units; this will permit the truss to be drawn correctly to a scale which is compatible with the PC-2's printer-plotter.

```

208 D=90/(B+A):S=D*S:IF S > 400 LET
    D=400/S*D:S=400
209 A=D*A:B=D*B:LF 8
210 GRAPH:GLCURSOR (80,0):SORGN

```

Printer-plotter draws free body diagram of truss.

```

215 COLOR 1
220 LINE -(0,-S)-(B,-S)-(A,0)-(0,0)
230 FOR I=1 TO M+1
235 F=D*X(I):G=D*Y(I)
240 LINE (0,-F)-(G,-F)
250 NEXT I
260 FOR I=1 TO J
270 F=D*X(I):G=D*Y(I):H=D*X(I+1)
280 LINE (0,-H)-(G,-F)
290 NEXT I
295 X(M+2)=S/D:Y(M+2)=B/D
300 K=M+2:P=J+2
310 FOR I=K TO P STEP -1
320 F=D*X(I):G=D*Y(I):H=D*X(I-1)
330 LINE (G,-F)-(0,-H)
340 NEXT I

```

In lines 350 and 355 the reaction arrows are drawn.

```

350 GLCURSOR (0,0):GOSUB 2000
355 GLCURSOR (0,-S):GOSUB 2000

```

Lines 360-380 draw load arrows for bottom chord.

```
360 FOR I=2 TO M+1
370 F=D*X(I):GLCURSOR (0,-F): GOSUB
    3000
380 NEXT I
```

Lines 390-430 draw load arrows for top chord.

```
390 FOR I=2 TO M+1
400 F=D*X(I):G=D*Y(I)
410 GLCURSOR (G,-F)
420 GOSUB 4000
430 NEXT I
435 GLCURSOR (-50,0):SORGN
```

Lines 440 prints title.

```
440 COLOR 2:ROTATE 1:LPRINT "FREE
    BODY DIAG-PRATT TRUSS"
450 G=50+1.5*A:GLCURSOR (G,100):
    SORGN:COLOR 0
```

Lines 460-500 identify panel numbers.

```
460 ROTATE 1:LPRINT "PANEL #":
    GLCURSOR (0,-70):SORGN
480 FOR I=1 TO (M+1)
490 F=(X(I+1)-X(I))/3+X(I):F=D*F:
    GLCURSOR (0,-F):USING
500 ROTATE 1:CSIZE 3:LPRINT I:NEXT I
```

Lines 510-523 identify left and right reactions, and the span.

```
510 G=G-15:GLCURSOR (-G,-10):
    SORGN:ROTATE 1:CSIZE 2:LPRINT
    "RL":F=S+25
520 GLCURSOR (0,-F):SORGN: ROTATE
    1:LPRINT "RR"
523 GLCURSOR (15,(0.7*F)):SORGN:
    ROTATE 1:LPRINT "SPAN=": INT
    (S/D*1000)/1000
525 ROTATE 0:TEXT:LF 15
```

Program requires user to input truss loads.

```
530 LPRINT:LPRINT "INPUT TOTAL
    LOADS":LPRINT "TOP, THEN
    BOTTOM":LPRINT "LEFT TO RIGHT,
    ON"
```

```
540 LPRINT "INTERIOR PANEL POINTS"
545 O=2:N=0
550 FOR I=2 TO (2*M) STEP 2: O=O+2:
    N=N+1:V=I+1:USING
555 LPRINT:LPRINT "PANEL PT "; (N+1):
    USING "#####.##"
560 LPRINT "TOP LOAD=":INPUT W(V):
    LPRINT W(V):LPRINT "BOT LOAD=":
    INPUT W(O):LPRINT W(O)
565 Q=Q+X(N+1)*W (V)+W(O)*X(N+1):
    R=R+W(V)+W(O)
570 NEXT I
```

Program calculates and prints left and right reactions.

```
571 RR=INT (Q/S*D*100+.5)/100:RL=INT
    ((R-RR)*100+.5)/100
572 LPRINT:LPRINT:LPRINT
    "***REACTIONS***"
575 LPRINT USING "#####.##": RL=
    ";RL:LPRINT " RR= ";RR
577 N=0:GAM (0)=0
580 J=(J-1)*10+1
587 W(2)=-RL:T(7)=0:W(1)=0:V=0
590 T(2)=0:T(3)=0:LF 7
```

Title "Member Forces" is printed.

```
592 GRAPH:ROTATE 1: COLOR 1:
    GLCURSOR (200,0):LPRINT "MEMBER
    FORCES":GLCURSOR (170,0):LPRINT
    "+=COMP; - = TENS":TEXT
595 LF 5
```

Program calculates member forces in each "up-left" panel.

```
600 FOR I=1 TO J STEP 10
610 T(I)=W(V+2)
630 T(I+3)=T(I+2)*SIN GAM(N)-T(I):
    T(I+3)=INT (100*T(I+3)+.5)/100
640 T(I+4)=T(I+1)+T(I+2)*COS GAM (N):
    T(I+4)=INT (T(I+4)*100+.5)/100
670 T(I+11)=T(I+4)
680 T(I+7)=W(V+1)
690 T(I+5)=T(I+3):N=N+1:V=V+2
700 T(I+9)=(T(I+5)-T(I+7))/(COS GAM (N)*
    TAN PHI+SIN GAM(N))
705 T(I+8)=T(I+6)+T(I+9)*COS GAM (N)/
    COS PHI
710 T(I+9)=INT (T(I+9)*100+.5)/100
```

```
715 T(I+8)=INT (T(I+8)*100+.5)/100
```

Program draws up-left panels and represents the member forces.

```
720 GOSUB 5990
740 T(I+16)=T(I+8):T(I+12)=T(I+9): NEXT
  I
750 K=10*(M+1)+1:T(K+1)=0:T(K+2)=0:
  T(K+6)=0:N=M+1:V=(M+2)*2:W(V)=-RR
760 W(V-1)=0:GAM (M+2)=0
770 GRAPH:COLOR 1:USING:GLCURSOR
  (80,0):SORGN:LINE -(100,0)
775 GLCURSOR (-30,60):ROTATE 1:
  LPRINT "VERT WEB MEM"
780 GLCURSOR (50,60):COLOR 3: USING
  "#####.###":LPRINT W(JJ):TEXT:
  COLOR 1:LF 5
```

Program calculates member force in up-right panels.

```
800 FOR I=K TO (J+20)STEP -10
810 T(I)=W(V)
830 T(I+3)=T(I+2)* SIN GAM (N+1)-T(I)
835 T(I+3)=INT (T(I+3)*100+.5)/100
840 T(I+4)=T(I+1)+T(I+2)*COS GAM (N+1):
  TEXT
850 T(I+4)=INT (T(I+4)*100+.5)/100
870 T(I-9)=T(I+4)
880 T(I+7)=W(V-1)
890 T(I+5)=T(I+3):V=V-2
900 T(I+9)=(T(I+5)-T(I+7))/(COS GAM
  (N) * TAN PHI+SIN GAM (N))
905 T(I+9)=INT (T(I+9)*100+.5)/100
910 T(I+8)=T(I+6)+T(I+9)*COS GAM (N)/
  COS PHI
920 T(I+8)=INT (T(I+8)*100+.5)/100
940 T(I-4)=T(I+8):T(I-8)=T(I+9):N=N-1:
  NEXT I
945 N=JP+1
```

Program draws up-right panels and represents member forces.

```
950 FOR I=(J+20) TO KSTEP 10: GOSUB
  6100
960 N=N+1:NEXT I
999 LF 4:COLOR 0:END
```

Subroutines 2000, 3000 and 4000 draw arrows.

```
2000 RLINE (-30,0)-(-30,0)-(-5,5)-(-10,
  -5,5):RETURN
3000 RLINE (-30,0)-(-5,-5)-(-10,10)- (-5,-5)
  -(-25,0):RETURN
4000 RLINE (30,0)-(-30,0)-(-5,-5)-(-10,10)
  -(-5,-5):RETURN
```

Subroutine 5000 calculates angle of web member, length of web diagonal, and length of top chord.

```
5000 GAM(I)=ATN (U/V):Z= V/(U^2+V^2):
  ZZ=V/COS PHI
5005 GAM (I)=INT (100*GAM(I)+.5)/100:
  Z=INT (100*Z+.5)/100:ZZ=INT (100*
  ZZ+.5)/100
5007 IF B$="N" THEN 5050
5010 LPRINT USING; "PANEL ";I:
  LPRINT "WEB ANGLE W/ HORIZ":
  LPRINT "=";USING "#####.###";
  GAM (I)
5015 LPRINT "LGTH,WEB DIAG":
  LPRINT "=";Z
5025 LPRINT "LGTH, TOP CHORD":
  LPRINT "=";ZZ
5030 LPRINT "LGTH BOT CHORD":
  LPRINT "=";V
5040 LPRINT "LGTH, VERT MEM":
  LPRINT "=";U: LPRINT
5050 RETURN
```

Subroutine 5990 draws up-left panel and represents member forces.

```
5990 GRAPH:USING:COLOR 1
6007 GLCURSOR (80,0):SORGN
6010 LINE -(0,100)-(-100,100)-(-100,0)
6020 LINE (100,100)-(-10,10):GLCURSOR
  (-10,90):SORGN
6030 GLCURSOR (-25,0):ROTATE 1:
  LPRINT "PANEL #";N:COLOR
  3:USING "#####.###"
6040 GLCURSOR (15,10):LPRINT -T(I+4):
  GLCURSOR (70,0):LPRINT -T(I+9):
  GLCURSOR (55,75):LPRINT T(I+5)
6043 GLCURSOR (115,10):LPRINT T(I+8)
6045 TEXT:LF 8:GRAPH
6050 RETURN
```

Subroutine 6100 draws up-right panel and represents member forces.

```

6100 GRAPH:USING:COLOR 1
6110 GLCURSOR (80,0):SORGN
6120 LINE -(0,-100)-(100,-100)-(100,0)
6130 GLCURSOR (100,-100)
6140 LINE -(15,-15):GLCURSOR (-10,0):
      SORGN
6150 GLCURSOR (-25,0):ROTATE 1
6160 LPRINT "PANEL #";N:COLOR 3:
      USING "#####.###"
6170 GLCURSOR (15,10):LPRINT -T(I+4)
6180 GLCURSOR (75,20):LPRINT -T(I+9)
6190 GLCURSOR (115,10):LPRINT T(I+8)
6195 GLCURSOR (55,-40):LPRINT T(I+5)
6200 TEXT:LF 1:RETURN

```

GCP PROGRAM - RECTANGULAR WOOD MEMBERS-COMBINED AXIAL COMPRES-SION AND FLEXURAL LOADING - X AND Y DIRECTIONS - SIDE LOADING AND ECCENTRICITY - PC-1/PC-2


Ref: "National Design Specification for Wood Construction," 1977 Edition, National Forest Products Association, Washington, D.C.

This combined axial and bending stress type of problem is a trial-and-error design process, lengthy and time consuming without the aid of a computer.

If a truss compression chord, 2 x 4 or smaller, (in combined axial load and flexural bending) is subjected to bending such that the edge to which the plywood is attached is in flexural compression, and if the truss is not used in "wet service conditions", and if the wood is seasoned to 19% moisture content or less, use C_T (buckling stiffness factor):

$$C_T = 1 + .002 * l_e$$

where l_e = effective buckling length used in the design of the chord for compressive loading, inches. For l_e values greater than 96", C_T is calculated using $l_e = 96$ "



References herein to the x-x axis and the y-y axis are intended to pertain to the "strong" way and the "weak" way, resp., of a rectangular section. See diagram.

Parameter Designations:

Input:

- I = Actual Member Depth (greater dimn), in.
- J = Actual Member Thickness, in.
- K = Unbraced Length, x-x, ft.
- L = Unbraced Length, y-y, ft.
- M = F_b = Allowable Stress, psi
- N = F_c = Allowable Compressive Stress, psi
- O = E = Modulus of Elasticity, psi
- P = Axial Load on Member, lb.
- Q = Load Factor, Transverse Load
- R = Load Factor, Axial Load
- S = Bending Moment, Max, x direction, ft-lb
- T = Bending Moment, Max, y direction, ft-lb
- U = Eccentricity, x-x, inches
- V = Eccentricity, y-y, inches
- W = End Restraint Coef., x-x axis
- X = End Restraint Coef., y-y axis

Determine whether member being designed is a truss compression chord, and whether member dimensions qualify for modification using the Buckling Stiffness Factor.

*See Errata
June Pg. 6*

```

100 WAIT 500: INPUT "TRUSS
    COMPR. CHORD? (Y/N)";H$:IF
    H$="N" THEN 112
110 H$="N":IF I<=4 AND J<=2
    LET H$="Y"
112 IF Q=0 LET Q=1
114 IF R=0 LET R=1
116 IF W=0 LET W=1
118 IF X=0 LET X=1
120 G=W*K*12/1:F=X*L*12/J:C=1
130 E=G:IF F>G LET E=F
140 IF E>50 PRINT "L/D>50:
    COLUMN INVALID":BEEP
    4:END

```

Calculate Buckling Stiffness Factor ($C_T = 1 + .002 * l_e$) It is assumed that, for this case, the plywood will be fastened to the narrow edge of the chord; no lateral bending will be involved. Therefore the bending moment in the y direction must equal zero.

```

150 IF T<>0 THEN 170
160 IF H$="Y" LET C=1+.002*12*K*W:
    IF (K*W)>8 LET C=1+.002*96

```

Calculate K and J values

```

170 A=.671 * V (O/N): A=A* V C
180 G=(G-11)/(A-11)
183 IF G<0 LET G=0
186 IF G>1 LET G=1
190 F=(F-11)/(A-11)
193 IF F<0 LET F=0
196 IF F>1 LET F=1

```

Calculate F_c

```

200 IF E>=A LET Y=.3*O/E/E
210 IF E>11 AND E<A LET Y=N*(1-
    (E/A)^4/3)
215 IF Y=0 LET Y=N

```

Calculate f_c and Interaction Number for Compression modified by Axial Load Factor

```
220 B=P/I/J/R/Y
```

Modify Interaction Number for Bending in x and y directions

```

230 D=(S*12*6/J/I/Q+B*Y*(6+1.5 *G)
    *U/I)/(M-G*B*Y)
240 D=D+(T*12*6/I/J/J/Q+B*Y*(6 +1.5
    *F)*V/J)/(M-F*B*Y)
250 A=B+D

```

We offer Cassette Tapes with documentation at \$20 each for:

- 1) Statistics-Confidence Program, PC-1 or PC-2
- 2) General Computation Program with or without printer, PC-1 or PC-2
- 3) Wood Column; Wood Foundation Wall; Wood Members - Combined Axial and Bending - PC-1 or PC-2
- 4) Steel Column and Steel Beam-Column, PC-2
- 5) Hazen-Williams and Chezy Manning Flow Formulas, PC-1 or PC-2
- 6) Walls Retaining Drained Sand, Active Case, PC-1 or PC-2
- 7) Pratt Truss, PC-2 Requires 8k module
- 8) Accounts Receivable, PC-2. This program prints detailed invoices, including aging. Prepares summary of business during the time period. Suitable for up to five hourly categories of charge rates. Requires 8K module.

Note: Civil Engineers Pocket Computer Monthly supports Radio Shack's PC-1 and PC-2 (Sharp PC-1211 and PC-1500). We believe our software will be helpful to civil engineers who have other equipment.

The software provided in this issue is solely for educational and experimental purposes. It is supplied "as-is" without warranty of any kind. We do not assume any liability for any direct, indirect, incidental or consequential damages relating to the use or application of the programs or information contained herein.

WORKED OUT EXAMPLES

In this issue we will provide design examples for the superstructure of a wood building

PRATT TRUSS

*****PRATT*****
*****TRUSS*****

BY H.C. HALL, PE

PGM SOLVES PRATT
TRUSS; MEMBER GEOM/
LGTH/FORCES; DRAWS
FB DIAG TO SCALE.

X AXIS: BOT CHORD
Y AXIS: LEFT REACT

TOP&BOT CHD NEED
NOT BE PARALLEL.

INPUT TRUSS DIMNS
BUT BE CONSISTENT
FOR EXAMPLE, INPUT
FEET OR INCHES
NOT BOTH

DESIGNATE
NO. OF PANELS
COUNTING FROM LEFT
WHICH ARE TO HAVE
UP-LEFT, DOWN-RIGHT
WEB MEMBERS

3
30.00
2.50
2.66
6
5.00
5.00
5.00
5.00
5.00
5.00

30.00

The user has the option of printing out a set of instructions.

The program provides a complete analysis of truss geometry, as well as analysis up to the point where the individual web and chord members are to be selected.

The program draws a free body diagram of the truss to scale.

Member forces are represented on a sketch of the panel; each panel is drawn and the member forces superimposed thereon.

Note: Dimensions must be in consistent units. In this example, all units are feet.

The program will design a 30 ft. truss consisting of six equal panels.

1. The program documents all input.

No. of "up-left" panels

Span

Depth at left end

Depth at right end

Total Number of panels

Width of Panel #1

Width of Panel #2

Width of Panel #3

Width of Panel #4

Width of Panel #5

Width of Panel #6

Sum of Panel Widths = Span

2. At this point, the program asks the user whether truss dimensions are to be documented. User will answer "yes" if all details of the truss' panel dimensions are desired; this procedure is somewhat time-consuming, and the user may not wish to have all these dimensions printed out each time the program is run.

ANGLE OF TOP CHORD
W/HORIZ= 0.31
DEG
SPAN= 30.00

TRUSS DEPTH AT LEF
T END= 2.50

TRUSS DEPTH AT RIG
HT END= 2.66

PANEL DIMNS

PANEL 1
WEB ANGLE W/HORIZ
= 26.56
LGTH:WEB DIAG
= 5.59
LGTH, TOP CHORD
= 5.00
LGTH, BOT CHORD
= 5.00
LGTH, VERT MEM
= 2.50

PANEL 2
WEB ANGLE W/HORIZ
= 26.83
LGTH:WEB DIAG
= 5.60
LGTH, TOP CHORD
= 5.00
LGTH, BOT CHORD
= 5.00
LGTH, VERT MEM
= 2.53


PANEL 3
WEB ANGLE W/HORIZ
= 27.11
LGTH:WEB DIAG
= 5.62
LGTH, TOP CHORD
= 5.00
LGTH, BOT CHORD
= 5.00
LGTH, VERT MEM
= 2.56

PANEL 4
WEB ANGLE W/HORIZ
= 27.56
LGTH:WEB DIAG
= 5.64
LGTH, TOP CHORD
= 5.00
LGTH, BOT CHORD
= 5.00
LGTH, VERT MEM
= 2.61

PANEL 5
WEB ANGLE W/HORIZ
= 27.83
LGTH:WEB DIAG
= 5.65
LGTH, TOP CHORD
= 5.00
LGTH, BOT CHORD
= 5.00
LGTH, VERT MEM
= 2.64

PANEL 6
WEB ANGLE W/HORIZ
= 28.07
LGTH:WEB DIAG
= 5.67
LGTH, TOP CHORD
= 5.00
LGTH, BOT CHORD
= 5.00
LGTH, VERT MEM
= 2.66

3. The free body diagram is drawn correctly to scale. Note that the program was written for a "parallel chord" truss. Although the chords need not be truly parallel, the program for drawing the free-body diagram will not necessarily accomodate a truss with a large angle between the top and bottom chords.



4. At this point the loads on top and bottom chords are input. Note that the numbers of the panel points are related to the panel number as follows: Panel Point #2 is to the left side of Panel #2, and so on.

INPUT TOTAL LOADS:
TOP, THEN BOTTOM
LEFT TO RIGHT, ON
INTERIOR PANEL POINTS

PANEL PT 2
TOP LOAD=
BOT LOAD= 370.00
100.00

PANEL PT 3
TOP LOAD= 370.00
BOT LOAD= 100.00

PANEL PT 4
TOP LOAD=
BOT LOAD= 370.00
100.00

PANEL PT 5
TOP LOAD= 370.00
BOT LOAD= 100.00

PANEL PT 6
TOP LOAD= 370.00
BOT LOAD= 100.00

5. The program prints out the reactions:

****REACTIONS****


RL= 1175.00

RR= 1175.00


6. The program now provides a sketch of each panel with member forces superimposed thereon.

MEMBER FORCES


+=COMP; -=TENS




PANEL # 1




PANEL # 2




PANEL # 3




VERT WEB MEM



PANEL # 4



PANEL # 5



PANEL # 6

This completes the analysis of the Pratt Truss. However, individual members must be analyzed.

ANALYSIS OF TOP CHORD

The enclosed GCP program designs the truss top chord for the critical area, Panels 3 and 4. It is found that a 2 x 4 is approved (Interaction Number = 0.99) for #2 Southern Pine, Kiln Dried, for a maximum concentric load of 4200 lb. and a maximum moment, x direction, of 185 ft-lb. Note that, for this top chord, a continuous member is assumed, i.e., no butt joints at Panel Points 3, 4, or 5. Thus an end coefficient of 0.65 may be used. Based on this analysis, 2 x 4 members may be used throughout the truss.

TRUSS COMP CHORD

** ** *

MEMBER D, IN = 3.5
 MEMBER W, IN = 1.5
 UNBR.L, X, FT = 4.3
 UNBR.L, Y, FT = 0
 FB, PSI = 1500
 FC, PSI = 1200
 E, PSI = 1600000
 AX. LOAD, LB = 4200
 LD. FACT, TR = 1.15
 LD. FACT, AX = 1.15
 MOM, X, FT-LB = 185
 MOM, Y, FT-LB = 0
 ECC., X, IN. = 0
 ECC., Y, IN. = 0
 END COEF., X = ~~0.65~~ 1
 END COEF., Y = 1

COMP. RESULTS:

INTERACT. NO = ~~9.999488453E-01~~

1.0776

ANALYSIS OF GLULAM BEAM

Check the design of a 6-3/4x21 glulam wood beam using Southern Pine, (Combination 16F-E1) carrying the wood Pratt trusses on a 30 ft. span, each side. Total load = 47 psf roof loading. Span = 20 ft.

GCP PROGRAM

I= beam width, in.

J= beam depth, in.

K= load, lb/ft

L= span, ft.

M= E, psi

100 $A = K / 8 * L \wedge 2$

110 $B = I / 6 * J \wedge 2$

120 $C = 12 * A / B$

130 $D = 3 / 2 * K * L / 2 / J$

140 $E = B / 2 * J$

150 $F = 5 * K * L \wedge 4 * 12 \wedge 3 / 384 / M / E$

GLULAM BEAM

** ** ** **

BM. WIDTH, IN = 6.75

BM. DPTH, IN. = 21

LOAD, LB/FT. = 1410

SPAN, FT. = 20

E, PSI = 1500000

COMP. RESULTS:

MOM, FT-LB. = 70500

SEC. MOD, IN² = 496.125

BDG. STR, PSI = 1705.21542

SHR. STR, PSI = 149.2063492

M. INERT, IN³ = 5209.3125

DEFLECT, IN = 0.649605874


The calculated bending stress is 1705 psi. The duration factor for snow load is taken as 1.15. Therefore $1705 / 1.15 = 1483$ psi which allows ample margin for self weight.

The calculated shear stress is 149 psi which is much less than the allowable 200 psi.

The deflection is 0.65 inches, which is about $1 / 360 \times$ Span.

ANALYSIS OF POLE COLUMN

Check the design of a 4X6 treated wood pole (#2 Southern Pine, Kiln Dried) of 10 ft. height, 8 ft. spacing, located along an exterior column line. The column is embedded to a depth of 3'-6" and bears on a plain concrete footing pad, thus:



AXIAL LOAD $30/2 * 8 * 47 \text{ psf} = 5640 \text{ LB.}$

The moment at point A = $9/128 w l^2 =$

$$9/128 * 8 * 15 * 10^2 = 843.75 \text{ FT-LB}$$

The moment at point B = $1/8 w l^2 = 15 * 10^2 = 1500 \text{ FT-LB}$

We will assume, for purposes of analysis, that the unbraced height at point A is 10 ft., and at point B, 2 ft. Two analyses are therefore required. The design interaction number does not exceed 0.763; therefore, the design is satisfactory.

WOOD POLE
** ** ** **

MEMBER D, IN = 5.5
MEMBER W, IN = 3.5
UNB. HT, X, FT = 10
UNB. HT, Y, FT = 2
FB, PSI = 1300
FC, PSI = 1200
E, PSI = 1600000
AX. LOAD, LB = 5640
LD. FACT, TR = 1.3333

*See Pg. 7
July*

LD. FACT, AX = 1.15
MOM, X, FT-LB = 843.75
MOM, Y, FT-LB = 0
ECC., X, IN. = 0
ECC., Y, IN. = 0
END COEF., X = 0.8
END COEF., Y = 0.8

COMP. RESULTS:

INTERACT. NO = ~~5.975103861E-01~~
0.613

WOOD POLE

** ** *

MEMBER D, IN =

5.5

MEMBER W, IN =

3.5

UNB. HT, X, FT =

2

UNB. HT, Y, FT =

2

FB, PSI =

1300

FC, PSI =

1200

E, PSI =

1600000

AX. LOAD, LB =

5640

LD. FACT, TR =

1.3333

LD. FACT, AX =

1.15

MOM, X, FT-LB =

1500

MOM, Y, FT-LB =

0

ECC., X, IN. =

0

ECC., Y, IN. =

0

END COEF., X =

~~0.8~~

END COEF., Y =

~~0.8~~

COMP. RESULTS:

INTERACT. NO =

~~0.762823363~~

.8571

ANALYSIS OF JOISTS

Check the design of a 2 X 8 wood joist @ 24" centers, using #2 Southern Pine, KD on a 12 ft. span. Total load = 47 psf roof loading. Use the GCP Program on page 13.

GEN COM PGM
RESULTS

A= 1692.
B= 13.140625
C= 1545.131986= f_b
D= 77.79310345= f_v
E= 47.63476563
F= 5.754284635E-01= Δ

The calculated bending/shear stresses are less than the allowable values, considering the duration factor of 1.15.

The deflection is 0.57 inches, which is about 1/240 X Span.

